Maple- Approximation to the smallest positive solution Part of my first year university degree, I need to work with the Maple software.
I have this question.
Find an approximation to the smallest positive solution to
$$(-38/25)*sin(9/10*x)=(13/25)*exp(-16*x/25)$$
Correct to 10 sig.figs
No matter what I do, I don't seem to get the correct answer.
Initially, I got a negative answer but then I sent the bound of x from 0 to infinity. Still getting the wrong answer.
This is my best attempt so far. 
fsolve((-38/25)*sin(9/10*x)=(13/25)*exp(-16*x/25),x,x=0..infinity);
10.47244230
All help is greatly appreciated

restart;
expr1 := -38/25*sin(9/10*x):
expr2 := 13/25*exp(-16/25*x):

P1 := plot(expr1,x=0..10,color=blue):
P2 := plot(expr2,x=0..10,color=red):
plots:-display(P1,P2,view=-2..2);


The purpose of creating the plot is to allow you visually
discover a range for x between which the two curves intersect
only once.
For example you can see that they intersect only once
between x=0 and x=5.
Thus you can figure out, by visual inspection of the plot,
a suitable range to use when calling fsolve.
fsolve(expr1-expr2, x=0..5);

                      3.530355063

Here are some other ways,
RootFinding:-NextZero(unapply(expr1-expr2,x), 0.0);

                      3.530355062

Student:-Calculus1:-Roots(expr1-expr2, x=0..10, numeric);

               [3.530355063, 6.976944937]

min(%);

                      3.530355063

